

Memory Diagnostic -- Prosser's Variation

All numbers in this documentation are octal.

This diagnostic is just a variation of another diagnostic that is given in the section on assembly of the LD14. This diagnostic may be loaded at the beginning of any page. The diagnostic tests the locations in sequence. First it loads a pattern (actually an octal number) into a location and then immediately checks the location to see if it contains the correct pattern. Next it loads a zero into the same location and checks to see if the location contains a zero. It then repeats this test for the next location. At the end of one complete test of memory the diagnostic can be made to halt if an instruction is changed while loading. If an error is encountered, control is given to a subroutine which will first display the bad location in the accumulator and then stop. When CONT is depressed, the contents of the bad location will be displayed in the accumulator and once again the computer will halt. The diagnostic will resume if CONT is depressed once again.

The diagnostic starts at XX00 which is the beginning of any page. Location XX46 contains the beginning address of the test area and location XX47 contains the ending address of the test area. Location XX44 contains the test pattern which may be any octal number. If this is loaded in page zero BEGIN is usually set to 0050. For 1k END is usually set to 1777 and for 4k it is usually 7777. This diagnostic executes in a few seconds. Don't select BEGIN and END such that the diagnostic is in the test area.

<u>Location</u>	<u>Instruction</u>	<u>Label</u>	<u>Mnemonic</u>
XX00	7300	START	CLA + CLL
01	1246		TAD BEGIN
02	3245		DCA ADDR
03	1244		TAD PATTERN
04	3645		DCA * ADDR
05	1244		TAD PATTERN
06	7041		CIA
07	1645		TAD * ADDR
10	7440		SZA
11	4232		JMS ERROR
12	7300		CLA + CLL
13	3645		DCA * ADDR
14	7041		CIA
15	1645		TAD * ADDR
16	7440		SZA
17	4232		JMS ERROR
20	7300		CLA + CLL
21	1245		TAD ADDR
22	7041		CIA
23	1247		TAD END
24	7450		SNA
25	5242		JMP RESTART
26	7300		CLA + CLL
27	1245		TAD ADDR
30	7001		IAC
31	5202		JMP START + 2
32	XXXX	ERROR	RSV 1
33	7300		CLA + CLL
34	1245		TAD ADDR
35	7402		HLT
36	7300		CLA + CLL
37	1645		TAD * ADDR
40	7402		HLT
41	5632		JMP * ERROR

<u>Location</u>	<u>Instruction</u>	<u>Label</u>	<u>Mnemonic</u>
42	7000 or 7402	RESTART	NOP (or could change to HLT)
43	5200		JMP START
44	PATTERN	PATTERN	any octal number, often 7777
45	XXXX	ADDR	variable
46	BEGIN	BEGIN	see last paragraph
47	END	END	see last paragraph